

## Remote Control Car

### 1. Deskripsi

Remote Control Car adalah salah satu jenis robot beroda dengan kendali menggunakan sebuah remote. Transmisi data antara robot dan remote dapat menggunakan berbagai macam media baik melalui kabel, inframerah, Bluetooth, hingga wifi. Robot yang kita buat kali ini akan menggunakan Bluetooth sebagai media transmisi datanya.

### 2. Alat dan Bahan


#### a. Remote Control

- i. Arduino
- ii. Modul Joystick
- iii. Modul Bluetooth HC-05
- iv. Jumper


#### b. Robot RC

- i. Arduino
- ii. Modul Driver Motor L298N
- iii. Modul Bluetooth HC-05
- iv. Chassis RC Car
- v. Motor DC
- vi. Jumper

### 3. Rangkaian Elektronik



Gambar 1 Skema elektronika untuk robot RC Car



Gambar 2 Skema elektronik untuk remote control

#### 4. Persiapan Pada Modul Bluetooth

Pertama-tama kita akan menentukan mana modul Bluetooth yang akan menjadi **Master** dan mana yang akan menjadi **Slave**. Modul Bluetooth yang akan dijadikan sebagai remote control kita akan set sebagai Master sedangkan modul Bluetooth pada robot akan menjadi Slave. Sebelum itu silahkan masukkan kodingan berikut kepada masing-masing Arduino terlebih dahulu:

**\*\*jika menggunakan mode master, maka Bluetooth akan mengirim pesan kesemua perangkat secara broadcast**

```
#include <SoftwareSerial.h>

SoftwareSerial bluetooth(2, 3);
char data;

void setup()
{
  Serial.begin(9600);
  Serial.println("Enter AT commands:");
  bluetooth.begin(38400);
}
```

```


void loop()
{
  // JIKA ADA PESAN DARI MODUL BLUETOOTH
  if (bluetooth.available()){
    // MAKA PESAN TERSEBUT AKAN DITAMPILKAN PADA SERIAL MONITOR
    Serial.print(bluetooth.readString());
  }

  // JIKA ADA COMMAND DARI USER
  if (Serial.available()){

    // MAKA COMMAND AKAN DIKIRIMKAN KE BLUETOOTH
    data = Serial.read();
    bluetooth.write(data);
  }
}

```

Selanjutnya kita akan melakukan setting pada modul Bluetooth, silahkan buka serial monitor lalu gunakan mode **Both NL & CR**.



Tabel 1 AT Command Modul Bluetooth HC-05

No	Command	Response	Keterangan
1	AT	OK	Tes koneksi
2	AT+NAME?	+NAME:<name bluetooth>	Untuk melihat nama Bluetooth saat ini
3	AT+NAME=<name>	OK	<name> diisi dengan nama yang diinginkan
4	AT+ADDR?	+ADDR:<address>	Cek MAC Address, contoh 98D3:A1:F5BCE1
5	AT+CMODE?	+CMODE:<mode>	Cek mode koneksi ----- CMODE: 0 mode slave 1 mode master
6	AT+CMODE=<mode>	OK	SET mode koneksi
7	AT+BIND?	+BIND:<address>	Cek address bluetooth yang terkoneksi
8	AT+BIND= <address>	OK	Set address yang akan dikoneksikan contoh FCA8,9A,58D5

a. Setting Bluetooth As Slave

AT

AT+NAME=<isikan nama>

AT+ADDR?

AT+CMODE=0

b. Setting Bluetooth As Master

AT


AT+NAME=<isikan nama>

AT+ADDR?

AT+CMODE=0

AT+BIND=<masukan mac address>

5. Robot Car



Gambar 3 Algoritma RC Car



File robot\_rc.ino

```
#include <SoftwareSerial.h>
```

```
// MENENTUKAN INSTANCE UNTUK BLUETOOTH  
SoftwareSerial bluetooth(2, 3);
```

```
// INISIALISASI PIN UNTUK MOTOR DC
```

```

int ENA = 10;
int IN1 = 9;
int IN2 = 8;
int IN3 = 7;
int IN4 = 6;
int ENB = 5;

// VARIABLE UNTUK MENAMPUNG DATA DARI REMOTE CONTROL
char data;

void setup() {
    bluetooth.begin(9600);
    Serial.begin(9600);

    // INISIALISASI PIN-PIN MOTOR DC
    init_motor(ENA, IN1, IN2, IN3, IN4, ENB);
}

void loop() {

    // CEK APAKAH ADA SIGNAL YANG DIKIRIM OLEH TRANSMITTER/REMOTE
    if (bluetooth.available() > 0) {

        // JIKA ADA SIGNAL DARI TRANSMITTER/REMOTE BACA DATA YANG DIKIRIM
        data = bluetooth.read();

        // CETAK DATA YANG DITERIMA KE SERIAL MONITOR
        Serial.println(data);

        // KONTROL SETIAP DATA
        if (data == 'F') {
            go_forward(100, 100);
        } else if (data == 'B') {
            go_backward(100, 100);
        } else if (data == 'R') {
            turn_right(100);
        } else if (data == 'L') {
            turn_left(100);
        } else {
            stop();
        }
    }

    // MEMBERI JEDA 50 milisekon UNTUK ARDUINO
    delay(50);

    // JIKA TIDAK ADA SIGNAL DARI TRANSMITTER/REMOTE MAKA ROBOT BERHENTI
    stop();
}

```

```
}
```

File motor.ino

```
void init_motor(int ENA, int IN1, int IN2, int IN3, int IN4, int ENB){
    pinMode(ENA, OUTPUT);
    pinMode(IN1, OUTPUT);
    pinMode(IN2, OUTPUT);
    pinMode(IN3, OUTPUT);
    pinMode(IN4, OUTPUT);
    pinMode(ENB, OUTPUT);
}

void go_forward(int speed1, int speed2){
    analogWrite(ENA, speed1);
    analogWrite(ENB, speed2);

    digitalWrite(IN1, HIGH);
    digitalWrite(IN2, LOW);
    digitalWrite(IN3, LOW);
    digitalWrite(IN4, HIGH);
}

void go_backward(int speed1, int speed2){
    analogWrite(ENA, speed1);
    analogWrite(ENB, speed2);

    digitalWrite(IN1, LOW);
    digitalWrite(IN2, HIGH);
    digitalWrite(IN3, HIGH);
    digitalWrite(IN4, LOW);
}

void turn_right(int speed){

    analogWrite(ENA, speed);
    analogWrite(ENB, speed);

    digitalWrite(IN1, LOW);
    digitalWrite(IN2, HIGH);
    digitalWrite(IN3, LOW);
    digitalWrite(IN4, HIGH);
}

void turn_left(int speed){
```

```

    analogWrite(ENA, speed);
    analogWrite(ENB, speed);

    digitalWrite(IN1, HIGH);
    digitalWrite(IN2, LOW);
    digitalWrite(IN3, HIGH);
    digitalWrite(IN4, LOW);


}

void stop(){
    analogWrite(ENA, 0);
    analogWrite(ENB, 0);

    digitalWrite(IN1, LOW);
    digitalWrite(IN2, LOW);
    digitalWrite(IN3, LOW);
    digitalWrite(IN4, LOW);
}

```

## 6. Remote Control



Gambar 4 Algoritma Remote Control



File remote\_control.ino

```
#include <SoftwareSerial.h>

SoftwareSerial bluetooth(2, 3);

// MENENTUKAN BATAS IDLE SUMBU X
int BATAS_ATAS_X = 522 + 50;
int BATAS_BAWAH_X = 522 - 50;

// MENENTUKAN BATAS IDLE SUMBU Y
int BATAS_ATAS_Y = 508 + 50;
int BATAS_BAWAH_Y = 508 - 50;

// UNTUK MENAMPUNG PERINTAH
char data;

void setup() {
  bluetooth.begin(9600);
  Serial.begin(9600);
}

void loop() {

  char data = command();

  // JIKA DATA SELAIN 'S'
  if (data != 'S') {

    // MAKA KIRIMKAN DATA KE BLUETOOTH
    bluetooth.write(data);
    Serial.println(data);
  }

  delay(50);
}
```

File command.ino

```
char command() {
  int Y = analogRead(A0);
  int X = analogRead(A1);
```



```
if (Y < BATAS_BAWAH_Y) {  
    return 'F';  
} else if (Y > BATAS_ATAS_Y) {  
    return 'B';  
} else if (X < BATAS_BAWAH_X) {  
    return 'L';  
} else if (X > BATAS_ATAS_X) {  
    return 'R';  
} else {  
    return 'S';  
}  
}
```